Lab 1       M2A3: Lab: Working with Various Instructions Using DEBUG

1. Introduction
The purpose of this experiment is to introduce you to a powerful 80x86 assembly/machine language utility called DEBUG.
Preparing Your DEBUG Environment
If you have an older version of Windows (95, or 98) or Windows XP 32-bit, or Windows Vista 32-bit, or Windows 7 32-bit, then the DEBUG utility program is built into the operating system. You can prove this to yourself by opening a Command Prompt window (Start, Run, type debug into the run field, and click OK, or just click the Command Prompt icon). Once inside the Command Prompt window, type debug and press Enter. If you get an error then DEBUG does not exist on your Windows system.

If you have 64-bit versions of Windows XP, Vista, 7, or 8, then you need to install some special software to create a DEBUG environment. Read and follow the instructions in the following document:
· debughelp2.doc
The special software you need (DosBox) as well as the actual DEBUG program are saved in the following ZIP file: 

· debugfix.zip
The debughelp2.doc document explains how to unzip the files and get DEBUG running through DosBox.
Procedure
1. Read Appendix D, pages 560 through 575. Perform all the example DEBUG command exercises that are presented.
2. Here is the DEBUG program you need to enter and execute:
a
mov ax,0
mov bx,1234
mov cx,5678
mov dx,9abc
and ax,bx
or ax,cx
xor ax,dx
not ax

r
t 8
q
3. Run DEBUG and enter the program and commands line by line. The blank line after the not ax instruction is needed to get out of the “a” command. 

4. Examine the DEBUG output. The final value in AX should be 333B. This would be a good time to do an Alt Print Screen to capture a screen shot of the DEBUG window to paste into your lab writeup.
5. Quit DEBUG and then run it again.
6. After entering the “a” command, change the first instruction from
mov ax,0
to
mov ax,????
where ???? is your 4-digit birth date. For example, if you were born on July 15, your instruction would be mov ax,0715
Then enter the rest of the instructions and commands.
7. Use DEBUG to run the modified program. Do another Alt Print Screen to grab the new DEBUG results for your lab writeup.
ELEC202 Microprocesors
Running DEBUG in Microsoft Vista, Windows 7, and Windows 8
Microsoft has chosen to remove the DEBUG utility from several Windows 64-bit operating systems.

However, it is still possible to run DEBUG after following a simple process.

First, you need to install a utility program called DOSBox.

Second, you install a copy of DEBUG that runs inside DOSBox.

Installation

1. Download the ZIP file DEBUGFIX and open it. It contains two files as shown:

When you extract the files, just choose C:\ as the destination folder and both files will automatically be saved into the C:\DEBUG folder.

2. Run the DOSBox0.74-win32-installer.exe program. It will install DOSBox on your Windows Vista , 7, or 8 computer.
3. Run the DOSBox utility by double-clicking its desktop icon:

You will see the following screen:

4. Mount the DEBUG folder by typing mount c c:\debug and pressing Enter. 

Important Note: If you get an error and you typed the mount c c:\debug command in correctly, you may not have extracted the ZIP files correctly. For your instructor to be able to assist you, you will need to do a screen shot showing the files in the DEBUG folder.
FYI, there is a tutorial on the DosBox website if you want to take a deeper look. Visit www.dosbox.com for more information.

5. Switch to drive C by typing c: and pressing Enter:

6. Run DEBUG by typing debug and pressing Enter. Here you can see the results of DEBUG’s r command:

Now, just type commands into DEBUG one by one.

For example, here is the short program from Module 2 Lab 2A:
a

mov ax,0

mov bx,1234

mov cx,5678

mov dx,9abc

and ax,bx

or ax,cx

xor ax,dx

not ax

r

t 8

q

Type each line as shown and press Enter at the end of each line. The blank line after the not ax instruction is needed to get DEBUG out of the “a” command.

Here is the DOSBox DEBUG session using these commands:

Now just the t 8 and q commands are left:

You can always grab a screen shot of the DOSBox DEBUG window by pressing Alt Print Screen and then pasting from the clipboard into your lab document.

Here are all the DEBUG programs from the various labs for your reference:

First program from Module 3 Lab 3A
a

mov dx,1

mov ax,6b48

mov bx,00ba

div bx

mov bl,3c

div bl

r

t 6

q

Second program from Module 3 Lab 3A:

a 200

db 5

db 8

db 4

db 7

db 6

a 100

mov si,200

mov al,[si]

mov ah,0

mov bh,0

mov bl,[si+1]

add ax,bx

mov bl,[si+2]

add ax,bx

mov bl,[si+3]

add ax,bx

mov bl,[si+4]

add ax,bx

mov bl,5

div bl

r

t 0e

q

Third program from Module 3 Lab 3A:

a

mov al,20

cmp al,10

cmp al,20

r

t 3

q

Fourth program from Module 3 Lab 3A:

a

mov al,90

cmp al,80

cmp al,a0

r

t 3

q

Notes: 
· In all of these programs, remember that the blank line is needed to get out of the “a” command and be able to begin entering other DEBUG commands, such as “r” or “t” or whatever.
· Use Alt Print Screen to grab a screenshot of the DEBUG command window so you can paste your results into your lab writeup. You may want to do this before using the “q” command to quit DEBUG.

LAB 2      M2A4: Lab: Assembling and Linking

1. Introduction
The purpose of this experiment is to setup the MASM assembler and linker. This will allow you to write assembly language source files and create executable code, such as .EXE and .COM files.

Procedure
1. Download the ZIPped MASM file: MASM6.EXE (4.6 MB)
2. This program is self-extracting. When you run it, you will see the following dialog box:

The default drive is C:. You may change this, but do not enter a directory name. The files end up in a directory called MASM611 automatically. So, if you just click Unzip without doing anything else, the files will be unzipped to
C:\MASM611.

3. You will do all your work with MASM in the MASM611\BIN folder. This means you will create, edit, and save your .ASM files in the MASM611\BIN folder. You will run the ML assembler and linker from the MASM611\BIN folder. Your LST files will be saved by ML in the MASM611\BIN folder as well. 

Finally, you will run your .EXE file from the MASM611\BIN folder as well.

Important Note: If you are running a 64-bit version of Windows, you will need 

to use DosBox to run the ML program. The initial command will be

mount c c:\masm611\bin
4. Download test.asm . 

You may need to right-click and choose Save target as.. to save it. Save it in the MASM611\BIN folder.

5. Open a DOS window and navigate to MASM611\BIN where TEST.ASM is stored. In the DOS command window, enter: 

CD \MASM611\BIN. On 64-bit Windows, start DosBox and duplicate the commands shown in the two figures provided here.

6. Enter the following command to assemble TEST.ASM:

ML /Fl TEST.ASM
This will create TEST.LST (the list file), and TEST.EXE (the executable code). Important: After the /F is a little l, not the number 1. If you use a little F or a big L you will get an error. The LST and EXE files will be in the MASM611\BIN folder.

7. Run the TEST.EXE program to verify it has been created. Include a copy of the LST file in your lab report. Refer to the following figures to see how TEST.ASM is assembled and executed.

Caption: DosBox window showing initial commands.

Caption: DosBox window showing assembling and execution of TEST.ASM
8. Choose three programs from the book (they are saved in book-cd-files.zip ) and create EXE files from them. Execute the programs and describe the results

. Download copy of test.asm
;Program TEST.ASM: Testing the MASM installation.

;

      .MODEL SMALL

      .DATA

MSG   DB    'Way to go... MASM works!',13,10,'$'

      .CODE

      .STARTUP

      LEA   DX,MSG   ;set up pointer to greeting

      MOV   AH,9     ;display string function

      INT   21H      ;DOS call

      .EXIT

      END

                  LAW 3  M3A3:Lab: Calculations and Comparisons

If viewing this through the Assignment tool, click the title above to go to the Submissions area
Introduction
The purpose of this experiment is to see how typical mathematical calculations, as well as numeric and logical comparisons, are performed
Procedure
1. The sun is 93,000,000 miles from the Earth The speed of light is 186,000 miles/second. How long does it take a ray of light to reach the Earth? The correct answer is 8 minutes, 20 seconds.
Open a command window (or DosBox), start DEBUG, and enter these commands
a
movdx,1
movax,6b48
movbx,00ba
divbx
movbl,3c
divbl

r
t6
q

1…Remember that a screen shot showing your DEBUG results should be included in your lab report What is the final value in the AX register? Does it represent some combination of 8 and 20? Hint: Remember that DEBUG only works with hexadecimal numbers
2…. Explain how the calculations were performed  in SUN.SCR. For example, were the numbers 93,000,000 and 186,000 represented anywhere? If not, why not?

3. Open a command window (or DosBox), start DEBUG, and enter the following commands. What is the final value in AL? Explain why this is correct.
200
db5
db8
db4
db7
db6

a100
movsi,200
moval,[si]
movah,0
movbh,0
movbl,[si+1]
addax,bx
movbl,[si+2]
addax,bx
movbl,[si+3]
addax,bx
movbl,[si+4]
addax,bx
movbl,5
divbl

r
t0e
q

4. Open a command window (or DosBox), start DEBUG, and enter the following commands.

a
moval,20
cmpal,10
cmpal,20

r
t3
q

Discuss the relationship between the instructions and the results indicated by the flags..

5. Repeat step 4 using these commands

a
moval,90
cmpal,80
cmpal,a0
r
t3
q

6. Download the program COMPARES.ASM. Assemble, link, and run.

7….Choose four additional types of conditional jump instructions (see page 124), and add the appropriate procedures to test the new jump instructions. Your final program will show the results of eight comparisons.

8…Submit the final ASM file.

;Download copy of Program COMPARES.ASM: Compare two words of data different ways.

;

        .MODEL SMALL

        .DATA

NUM1    DW    25000

NUM2    DW    26000

AXMSG   DB    'AX equals$'

BXMSG   DB    'BX equals$'

MSG     DB    ' '

TENTHOS DB    ?

THOUS   DB    ?

HUNS    DB    ?

TENS    DB    ?

ONES    DB    ?

        DB    0DH,0AH,'$'

IMSG    DB    'Instruction: CMP AX,BX',13,10,'$'

JCTMSG  DB    'JC is True',13,10,'$'

JCFMSG  DB    'JC is False',13,10,'$'

JZTMSG  DB    'JZ is True',13,10,'$'

JZFMSG  DB    'JZ is False',13,10,'$'

JATMSG  DB    'JA is True',13,10,'$'

JAFMSG  DB    'JA is False',13,10,'$'

JLTMSG  DB    'JL is True',13,10,'$'

JLFMSG  DB    'JL is False',13,10,'$'

        .CODE

        .STARTUP

        LEA   DX,AXMSG          ;display AX value

        MOV   AH,9

        INT   21H

        MOV   AX,NUM1

        CALL  BTOD

        LEA   DX,MSG  

        MOV   AH,9

        INT   21H

        LEA   DX,BXMSG          ;display BX value

        MOV   AH,9

        INT   21H

        MOV   AX,NUM2           

        CALL  BTOD

        LEA   DX,MSG  

        MOV   AH,9

        INT   21H

        LEA   DX,IMSG           ;show 'CMP AX,BX'

        MOV   AH,9

        INT   21H

        MOV   AX,NUM1           ;start the comparisons

        MOV   BX,NUM2

        CALL  DOJC

        MOV   AX,NUM1

        MOV   BX,NUM2

        CALL  DOJZ

        MOV   AX,NUM1

        MOV   BX,NUM2

        CALL  DOJA

        MOV   AX,NUM1

        MOV   BX,NUM2

        CALL  DOJL

        .EXIT

DOJC    PROC  FAR

        CMP   AX,BX

        JC    JCOK

        LEA   DX,JCFMSG

        MOV   AH,9

        INT   21H

        RET

JCOK:   LEA   DX,JCTMSG

        MOV   AH,9

        INT   21H

        RET

DOJC    ENDP

DOJZ    PROC  FAR

        CMP   AX,BX

        JZ    JZOK

        LEA   DX,JZFMSG

        MOV   AH,9

        INT   21H

        RET

JZOK:   LEA   DX,JZTMSG

        MOV   AH,9

        INT   21H

        RET

DOJZ    ENDP

DOJA    PROC  FAR

        CMP   AX,BX

        JA    JAOK

        LEA   DX,JAFMSG

        MOV   AH,9

        INT   21H

        RET

JAOK:   LEA   DX,JATMSG

        MOV   AH,9

        INT   21H

        RET

DOJA    ENDP

DOJL    PROC  FAR

        CMP   AX,BX

        JL    JLOK

        LEA   DX,JLFMSG

        MOV   AH,9

        INT   21H

        RET

JLOK:   LEA   DX,JLTMSG

        MOV   AH,9

        INT   21H

        RET

DOJL    ENDP

BTOD    PROC  FAR

        SUB   DX,DX

        MOV   BX,10000

        DIV   BX                ;get 10,000s digit

        ADD   AL,30H

        MOV   TENTHOS,AL

        XCHG  AX,DX

        SUB   DX,DX

        MOV   BX,1000           ;get 1,000s digit

        DIV   BX

        ADD   AL,30H

        MOV   THOUS,AL

        XCHG  DX,AX

        MOV   BL,100

        DIV   BL          ;get hundreds digit

        ADD   AL,30H      ;convert into ASCII digit

        MOV   HUNS,AL     ;and save

        XCHG  AL,AH       ;get remainder

        SUB   AH,AH       ;prepare for division by 10

        MOV   BL,10

        DIV   BL          ;get tens digit

        ADD   AL,30H      ;convert into ASCII digit

        MOV   TENS,AL     ;and save

        ADD   AH,30H      ;convert ones digit into ASCII

        MOV   ONES,AH     ;and save

        RET

BTOD    ENDP

        END

LAB 4   M3A4: Lab: Loops and Subroutines

1…..If viewing this through the Assignment tool, click the title above to go to the Submissions area.
Introduction
The purpose of this experiment is to examine the operation of loops and their close equivalent, subroutines.

Procedure
1. How many times does each loop execute?

2. Download LOOPS.ASM. Assemble and link to create LOOPS.EXE. Run the program.

3. Comment out the PUSH AX and POP AX instructions. Re-assemble, link, and run. Comment on any changes. Explain how the changes, if any, occurred.

4. Put the PUSH AX and POP AX instructions back in.

5. Take instructions from inside the loop and place them in a subroutine (or procedure). Thus, the loop will call the subroutine once each pass. Only instructions that directly control the looping should stay.

6. Assemble, link, and run the program. You should get the same results. Email the final ASM file.

7. Download LOOPS2.ASM. Assemble and link to create LOOPS2.EXE. Run the program. Explain why the loop pass counter value shown is correct.

8. Download SUBS.ASM. This file contains errors! Edit the file to correct the errors, then assemble, link, and run. Warning! The program may assemble and link correctly and still have errors that may crash your DOS window. The correct result in AL is 7.

9. Submit your findings.

Downloaded copy of Program LOOPS.ASM: Count the number of passes through a loop.

;

        .MODEL SMALL

        .DATA

LOOPS   DW    ?

MSG     DB    'The number of loop passes is '

TENTHOS DB    ?

THOUS   DB    ?

HUNS    DB    ?

TENS    DB    ?

ONES    DB    ?

        DB    0DH,0AH,'$'

        .CODE

        .STARTUP

        MOV   AX,0      ;clear loop pass counter

;

;place user loop code here.

;Make sure INC  AX is inside the innermost loop.

;

        MOV   BX,0FF98H

BAK:    PUSH  AX

        MOV   CX,1234H

        MOV   DX,5678H

        XOR   CX,DX

        MOV   AX,9ABCH

        XOR   CX,AX

        POP   AX

        INC   AX                ;increment loop pass counter

        DEC   BX

        JNZ   BAK

;

;end of user loop code

;

        MOV   LOOPS,AX          ;display the loop pass counter

        CALL  BTOD

        LEA   DX,MSG  

        MOV   AH,9

        INT   21H

        .EXIT

BTOD    PROC  FAR

        MOV   AX,LOOPS    ;load binary input value

        SUB   DX,DX

        MOV   BX,10000

        DIV   BX                ;get 10,000s digit

        ADD   AL,30H

        MOV   TENTHOS,AL

        XCHG  AX,DX

        SUB   DX,DX

        MOV   BX,1000           ;get 1,000s digit

        DIV   BX

        ADD   AL,30H

        MOV   THOUS,AL

        XCHG  DX,AX

        MOV   BL,100

        DIV   BL          ;get hundreds digit

        ADD   AL,30H      ;convert into ASCII digit

        MOV   HUNS,AL     ;and save

        XCHG  AL,AH       ;get remainder

        SUB   AH,AH       ;prepare for division by 10

        MOV   BL,10

        DIV   BL          ;get tens digit

        ADD   AL,30H      ;convert into ASCII digit

        MOV   TENS,AL     ;and save

        ADD   AH,30H      ;convert ones digit into ASCII

        MOV   ONES,AH     ;and save

        RET

BTOD    ENDP

        END

LAB 5   M4A2: Lab: Interrupts

1. Introduction
The purpose of this experiment is to utilize the power of the 80x86 interrupt system.

Procedure
1. Open a DOS window and enter this command to start DEBUG and echo all its output to the file INTS.TXT:

DEBUG > INTS.TXT
2. You will not be able to see what you are typing, but enter this command next:

d 0:0 3ff
DEBUG will write a copy of the interrupt vector table to the INTS.TXT file.

3. Again, typing blind, enter a q to quit DEBUG.

4. Examine the INTS.TXT file and comment on anything interesting you discover. As an example, here is the INTS.TXT file created on my home computer.

5. Based on your INTS.TXT file, what are the CS:IP addresses for the following interrupts?
(a) INT 10H (Video BIOS)


(b) INT 13H (Disk Services)


(c) INT 21H (DOS Functions)

6. Examine Section 8.5 (pages 293 through 296). Then download VECTORS.ASM and assemble, link, and run.

7. To get a hardcopy of the VECTORS output, enter this DOS command:

VECTORS > VECS.TXT
You will need to hit 'Enter' four times to get back to the DOS prompt. Here is the sample VECS.TXT file from my computer.

8. Compare the INTS.TXT file with the results written to VECS.TXT.

9. Submit both TXT files with your comments.

;Download copy of Program VECTORS.ASM: View entire interrupt vector table.

;

        .MODEL SMALL

        .DATA

WMSG    DB    0DH,0AH,'Hit any key to continue...',0DH,0AH,0DH,0AH,'$'

        .CODE

        .STARTUP

        SUB   AL,AL       ;begin with vector 0

NEWV:   PUSH  AX          ;save vector number

        CALL  DISPHEX     ;display it

        MOV   DL,' '      ;load blank character

        MOV   AH,2        ;display character function

        INT   21H         ;DOS call

        MOV   DL,'='      ;load equal sign

        INT   21H         ;and output

        MOV   DL,' '      ;load another blank

        INT   21H         ;and output

        POP   AX          ;get vector number back

        PUSH  AX          ;and save it again

        MOV   AH,35H      ;get interrupt vector function

        INT   21H         ;DOS call

        MOV   AX,ES       ;get interupt segment

        CALL  DISPHEX_16  ;display it

        MOV   DL,':'      ;load colon character

        MOV   AH,2        ;display character function

        INT   21H         ;DOS call

        MOV   AX,BX       ;get interrupt offset

        CALL  DISPHEX_16  ;display it

        MOV   DL,' '      ;load blank character

        MOV   AH,2        ;display character function

        INT   21H         ;DOS call

        INT   21H         ;output a second blank

        POP   AX          ;get vector number back

        MOV   DL,AL       ;make a copy of it

        AND   DL,3        ;should we output a new line?

        CMP   DL,3

        JNZ   NNL

        PUSH  AX          ;save vector number

        MOV   DL,0DH      ;load carriage return

        MOV   AH,2        ;display character function

        INT   21H         ;DOS call

        MOV   DL,0AH      ;also output a line feed

        INT   21H

        POP   AX          ;get vector number back

NNL:    MOV   DL,AL       ;make copy of vector number

        CMP   DL,255      ;finished?

        JZ    NNP

        AND   DL,63       ;should we pause here?

        CMP   DL,63

        JNZ   NNP

        PUSH  AX          ;save vector number

        LEA   DX,WMSG     ;set up pointer to pause message

        MOV   AH,9        ;display string function

        INT   21H         ;DOS call

        MOV   AH,8        ;character input function

        INT   21H         ;DOS call

        POP   AX          ;get vector number back

NNP:    INC   AL          ;advance to nect vector

        CMP   AL,0        ;did we wrap from 255 to 0?

        JNZ   NEWV

        .EXIT

DISPHEX_16 PROC NEAR

     PUSH  AX             ;save number

     MOV   AL,AH          ;load upper byte



     CALL  DISPHEX        ;go display in hex

     POP   AX             ;get number back

     CALL  DISPHEX        ;go display lower byte

     RET

DISPHEX_16 ENDP

DISPHEX PROC NEAR

        PUSH AX           ;save number

        SHR  AL,1         ;get upper nybble

        SHR  AL,1

        SHR  AL,1

        SHR  AL,1

        CALL HEXOUT       ;display hex character

        POP  AX           ;get number back

        AND  AL,0FH       ;preserve lower nybble

        CALL HEXOUT       ;display hex character

        RET

DISPHEX ENDP

HEXOUT  PROC NEAR

        CMP  AL,10        ;is AL greater than 10?

        JC   NHA1         ;yes

        ADD  AL,7         ;no, add alpha bias

NHA1:   ADD  AL,30H       ;add ASCII bias

        MOV  DL,AL        ;load output character

        MOV  AH,2         ;display character function

        INT  21H          ;DOS call

        RET

HEXOUT  ENDP

        END

You will need to hit 'Enter' four times to get back to the DOS prompt. Here is the sample VECS.TXT file from my computer
00 = 00C9:0F9E  01 = 0070:0465  02 = 0669:0016  03 = 0070:0465  

04 = 0070:0465  05 = F000:FF54  06 = F000:7E08  07 = F000:E7D0  

08 = 094B:0000  09 = 0669:0028  0A = F000:EF6F  0B = F000:EF6F  

0C = F000:EF6F  0D = F000:EF6F  0E = 0669:009A  0F = 0070:0465  

10 = 09F3:000B  11 = F000:F84D  12 = F000:F841  13 = FD6F:2427  

14 = F000:E739  15 = 0271:053A  16 = 0070:042D  17 = 0439:0A28  

18 = F000:E7A4  19 = 072A:002F  1A = F000:FE6E  1B = 0439:0604  

1C = 094B:001D  1D = F000:F0A4  1E = 0000:0522  1F = C000:2041  

20 = 00C9:0FA8  21 = 0864:0445  22 = FDDD:1D47  23 = 0E12:016D  

24 = 0E12:0178  25 = 00C9:0FBC  26 = 0217:0037  27 = 00C9:0FD0  

28 = 00C9:106C  29 = 0070:0466  2A = 0439:05B4  2B = 00C9:106C  

2C = 00C9:106C  2D = 00C9:106C  2E = 0E12:0162  2F = 0E13:01CC  

30 = C90F:E4EA  31 = F000:E700  32 = 00C9:106C  33 = 0945:0001  

34 = 00C9:106C  35 = 00C9:106C  36 = 00C9:106C  37 = 00C9:106C  

38 = 00C9:106C  39 = 00C9:106C  3A = 00C9:106C  3B = 00C9:106C  

3C = 00C9:106C  3D = 00C9:106C  3E = 00C9:106C  3F = 00C9:106C  

Hit any key to continue...

40 = F000:EC59  41 = F000:EFF5  42 = F000:F065  43 = C000:2454  

44 = F000:E7D0  45 = F000:E7D0  46 = F000:E7D0  47 = F000:E7D0  

48 = F000:E7D0  49 = F000:E7D0  4A = F000:E7D0  4B = FD4F:2627  

4C = F000:E7D0  4D = F000:E7D0  4E = F000:E7D0  4F = 0070:04FC  

50 = F000:EF6F  51 = F000:E7D0  52 = F000:E7D0  53 = F000:E7D0  

54 = F000:E7D0  55 = F000:E7D0  56 = F000:E7D0  57 = F000:E7D0  

58 = F000:E7D0  59 = F000:E7D0  5A = F000:E7D0  5B = F000:E7D0  

5C = 0975:0003  5D = F000:E7D0  5E = F000:E7D0  5F = F000:E7D0  

60 = 0000:0000  61 = 0000:0000  62 = 0000:0000  63 = 0000:0000  

64 = 0000:0000  65 = 0000:0000  66 = 0000:0000  67 = 0946:0040  

68 = F000:E7D0  69 = F000:E7D0  6A = F000:E7D0  6B = F000:E7D0  

6C = F000:E7D0  6D = C000:131A  6E = F000:E7D0  6F = F000:E7D0  

70 = 0669:0035  71 = F000:ECF3  72 = F000:EF6F  73 = F000:EF6F  

74 = 0669:00E2  75 = F000:F0FC  76 = 0669:00FA  77 = 0669:0112  

78 = 0000:0000  79 = 0000:0000  7A = 0000:0000  7B = 0000:0000  

7C = 0000:0000  7D = 0000:0000  7E = 0000:0000  7F = 0000:0000  

Hit any key to continue...

80 = 0000:0000  81 = 0000:0000  82 = 0000:0000  83 = 0000:0000  

84 = 0000:0000  85 = 0000:0000  86 = 0000:0000  87 = 0000:0000  

88 = 0000:0000  89 = 0000:0000  8A = 0000:0000  8B = 0000:0000  

8C = 0000:0000  8D = 2EA7:0022  8E = 0202:FCC7  8F = F000:FFFF  

90 = 0000:0000  91 = 0000:0000  92 = 0000:0000  93 = 0000:0000  

94 = 0000:0000  95 = 0000:0000  96 = 0000:0000  97 = 0000:0000  

98 = 0000:0000  99 = 0000:0000  9A = 0000:0000  9B = 0000:0000  

9C = 0000:0000  9D = 0000:0000  9E = 0000:0000  9F = 0000:0000  

A0 = 0000:0000  A1 = 0000:0000  A2 = 0000:0000  A3 = 0000:0000  

A4 = 0000:0000  A5 = 0000:0000  A6 = 0000:0000  A7 = 0000:0000  

A8 = 0000:0000  A9 = 0000:0000  AA = 0000:0000  AB = 0000:0000  

AC = 0000:0000  AD = 0000:0000  AE = 0000:0000  AF = 0000:0000  

B0 = 0000:0000  B1 = 0000:0000  B2 = 0000:0000  B3 = 0000:0000  

B4 = 0000:0000  B5 = 0000:0000  B6 = 0000:0000  B7 = 0000:0000  

B8 = 0000:0000  B9 = 0000:0000  BA = 0000:0000  BB = 0000:0000  

BC = 0000:0000  BD = 0000:0000  BE = 0000:0000  BF = 0000:0000  

Hit any key to continue...

C0 = 0000:0000  C1 = 0000:0000  C2 = 0000:0000  C3 = 0000:0000  

C4 = 0000:0000  C5 = 0000:0000  C6 = 0000:0000  C7 = 0000:0000  

C8 = 0000:0000  C9 = 0000:0000  CA = 0000:0000  CB = 0000:0000  

CC = 0000:0000  CD = 0000:0000  CE = 0000:0000  CF = 0000:0000  

D0 = 0000:0000  D1 = 0000:0000  D2 = 0000:0000  D3 = 0000:0000  

D4 = 0000:0000  D5 = 0000:0000  D6 = 0000:0000  D7 = 0000:0000  

D8 = 0000:0000  D9 = 0000:0000  DA = 0000:0000  DB = 0000:0000  

DC = 0000:0000  DD = 0000:0000  DE = 0000:0000  DF = 0000:0000  

E0 = 0000:0000  E1 = 0000:0000  E2 = 0000:0000  E3 = 0000:0000  

E4 = 0100:0000  E5 = 00EF:6D01  E6 = 3C00:82F0  E7 = 01ED:4C80  

E8 = DF93:D800  E9 = 0A02:DF02  EA = 2205:0A05  EB = 2205:0AED  

EC = 86F0:00ED  ED = 56C0:2C00  EE = 46F0:00ED  EF = CEA1:BF02  

F0 = 0007:C0A1  F1 = 8000:8000  F2 = 0000:0100  F3 = 007C:0000  

F4 = 0002:0100  F5 = 1080:0280  F6 = 0203:00C4  F7 = 0000:0E00  

F8 = 0001:7C00  F9 = 0001:0080  FA = 0000:0000  FB = 0001:0003  

FC = 0001:A04F  FD = 57E3:7C00  FE = 524A:5363  FF = 0246:F000  
1. M5A3: Lab: Controlling the PC

2. Instructions 

If viewing this through the Assignment tool, click the title above to go to the Submissions area.
Introduction
The purpose of this experiment is to directly control the hardware of the PC. To prevent damage to sensitive hardware, we will only play with the hardware timer that controls the speaker

Procedure
1. Download the program TONE.ASM. Assemble, link, and run. Hopefully, you will hear the speaker beep at a 1 KHz rate for a short period of time.

2. Adjust the INNER and OUTER values until the tone lasts for 5 seconds. How many loops are required? What is the speed of the processor on your system?

3. Re-adjust the INNER and OUTER values for a tone duration of 0.25 seconds.

4. Modify the TONE.ASM program so that the 0.25 second 1 KHz tone plays five times, once each second. Hint: Use DOS INT 21H Function 2CH (Get Time) to read the system clock. Look for the seconds value to change. When it does, play the 0.25 second tone, then start watching the system clock again, waiting for the seconds to change again. Do this five times.

Here is some sample code for detecting when a new second has started:

[image: image15.png]MoV AH, 2CH
InT 21H

The loop exits at the beginning of the new second. You will need to define NOW in the DATA portion of the program (NOW DB ?).

5. Submit the final ASM file.

;download copy of Program TONE.ASM: Generate a 1 KHz tone on the PC speaker for 5 seconds.

;

        .MODEL SMALL

        .DATA

;Note: You will have to adjust these values to get the 5 seconds.

;The current values give just over 1 second on a 1.3 GHz machine.

OUTER   DW    4000            ;outer loop count

INNER   DW    50000           ;inner loop count

        .CODE

        .STARTUP

        CALL  SPKRON          ;turn speaker on

        MOV   CX,1190         ;divisor for 1 KHz tone

        CALL  LDTIMER         ;set speaker frequency

        CALL  DELAY           ;wait for chosen duration

        CALL  SPKROFF         ;turn speaker off

        .EXIT

SPKRON  PROC  NEAR

        IN    AL,61H          ;read current state of port 61h

        OR    AL,3            ;set speaker control bits

        OUT   61H,AL          ;output new state

        RET

SPKRON  ENDP

SPKROFF PROC  NEAR

        IN    AL,61H          ;read current state of port 61h

        AND   AL,0FCH         ;clear speaker control bits

        OUT   61H,AL          ;output new state

        RET

SPKROFF ENDP

DELAY   PROC  NEAR

        MOV   DX,OUTER

WAIT1:  MOV   CX,INNER

WAIT2:  NOP

        NOP

        NOP

        NOP

        LOOP  WAIT2

        DEC   DX              

        JNZ   WAIT1           

        RET

DELAY   ENDP

;Note: Output frequency equals 1,190,000 divided by CX

LDTIMER PROC  NEAR

        MOV   AL,0B6H         ;timer 2 control word

        OUT   43H,AL

        MOV   AL,CL           ;output lower byte of count

        OUT   42H,AL

        MOV   AL,CH           ;output upper byte of count

        OUT   42H,AL

        RET

LDTIMER ENDP

        END

LAB 7 M5A4: Lab: Motherboards

1. Introduction
The purpose of this experiment is to examine the features of a typical PC motherboard, including:

· CPU and co-processor

· Memory

· I/O

· BIOS

· Interrupts

Procedure
1. Download EX8-14.PDF (6.7 MB) and give it a detailed skim.

2. Choose two chapters in the PDF for closer study. For each: 1) Do the "Questions/Activities" section and 2) Answer the questions in the Self Test section 

3. Submit your findings.
Download copy of Program TONE.ASM: Generate a 1 KHz tone on the PC speaker for 5 seconds.

;

        .MODEL SMALL

        .DATA

;Note: You will have to adjust these values to get the 5 seconds.

;The current values give just over 1 second on a 1.3 GHz machine.

OUTER   DW    4000            ;outer loop count

INNER   DW    50000           ;inner loop count

        .CODE

        .STARTUP

        CALL  SPKRON          ;turn speaker on

        MOV   CX,1190         ;divisor for 1 KHz tone

        CALL  LDTIMER         ;set speaker frequency

        CALL  DELAY           ;wait for chosen duration

        CALL  SPKROFF         ;turn speaker off

        .EXIT

SPKRON  PROC  NEAR

        IN    AL,61H          ;read current state of port 61h

        OR    AL,3            ;set speaker control bits

        OUT   61H,AL          ;output new state

        RET

SPKRON  ENDP

SPKROFF PROC  NEAR

        IN    AL,61H          ;read current state of port 61h

        AND   AL,0FCH         ;clear speaker control bits

        OUT   61H,AL          ;output new state

        RET

SPKROFF ENDP

DELAY   PROC  NEAR

        MOV   DX,OUTER

WAIT1:  MOV   CX,INNER

WAIT2:  NOP

        NOP

        NOP

        NOP

        LOOP  WAIT2

        DEC   DX              

        JNZ   WAIT1           

        RET

DELAY   ENDP

;Note: Output frequency equals 1,190,000 divided by CX

LDTIMER PROC  NEAR

        MOV   AL,0B6H         ;timer 2 control word

        OUT   43H,AL

        MOV   AL,CL           ;output lower byte of count

        OUT   42H,AL

        MOV   AL,CH           ;output upper byte of count

        OUT   42H,AL

        RET

LDTIMER ENDP

        END

LAB8   M6A2: Lab: The Floating-Point Unit

1. Introduction
The purpose of this experiment is to examine the operation of the floating-point unit.

Procedure
1. Using Example 11.19 as a guide, show how the number 355.6 is converted into a normalized short real format.

2. Modify the SQRT.ASM program so that the input number (FPX) is 355.6.

3. Assemble, link, and run the program. What are the results?

4. Does your hand conversion from Step 1 match the assemblers conversion in the list file?

5. Show how SQRT.ASM can be modified to display eight digits each for the integer and fractional parts of the converted number.

6. Do Programming Exercise 11.3.

7. Submit your findings.

